In[ ]:=

```
Clear["Global`*"]
```

Problem 1

In[ ]:=

```
Clear["Global`*"]
```

The density of a sphere of unit radius is proportional to 5 z2 + x y + 1 with the center of the sphere at the origin.Determine the elements of the moment of inertia tensor in this system.Specify the coordinate system in which the moment of inertia tensor is diagonal and find the elements of the tensor in this system.

In[ ]:=

```
xRule = {x, y, z} -> CoordinateTransform [ "Spherical " -> "Cartesian " , {r, θ, ϕ}] // Thread
```

Out[ ]= $\{x \to r\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta],\ y \to r\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[\phi],\ z \to r\,\mathrm{Cos}[\theta]\}$

In[ ]:=

```
rRule = {r, θ, ϕ} -> CoordinateTransform [ "Cartesian " → "Spherical ", {x, y, z}] // Thread
```

Out[ ]= $\left\{r \to \sqrt{x^2+y^2+z^2},\ \theta \to \mathrm{ArcTan}\left[z, \sqrt{x^2+y^2}\right],\ \phi \to \mathrm{ArcTan}[x, y]\right\}$

In[ ]:=

```
ρ = 5 z^2 + x y + 1;
ρ /. xRule
```

Out[ ]= $1 + 5\,r^2\,\mathrm{Cos}[\theta]^2 + r^2\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\theta]^2\,\mathrm{Sin}[\phi]$

In[ ]:=

```
mat = {{y^2 + z^2, - x y, - x z},
    {-y x, x^2 + z^2, -y z},
    {-z x, -z y, x^2 + y^2}};
mat // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} y^2+z^2 & -x\,y & -x\,z \\ -x\,y & x^2+z^2 & -y\,z \\ -x\,z & -y\,z & x^2+y^2 \end{pmatrix}$$

In[ ]:=

```
Integrate[ρ mat[[1, 1]] r^2 Sin[θ] /. xRule , {r, 0, 1}, {θ, 0, π}, {ϕ, 0, 2 π}]
```

Out[ ]= $\frac{136\,\pi}{105}$

In[ ]:=

```
mat2 = Integrate[ρ # r^2 Sin[θ] /. xRule , {r, 0, 1}, {θ, 0, π}, {ϕ, 0, 2 π}] & /@ mat
```

Out[ ]= $\left\{\left\{\frac{136\,\pi}{105}, -\frac{4\,\pi}{105}, 0\right\}, \left\{-\frac{4\,\pi}{105}, \frac{136\,\pi}{105}, 0\right\}, \left\{0, 0, \frac{32\,\pi}{35}\right\}\right\}$

In[ ]:=
```
mat2 // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} \frac{136\pi}{105} & -\frac{4\pi}{105} & 0 \\ -\frac{4\pi}{105} & \frac{136\pi}{105} & 0 \\ 0 & 0 & \frac{32\pi}{35} \end{pmatrix}$$

In[ ]:=
```
evec = (Normalize /@ Eigenvectors[mat2]) // Simplify // Transpose ;
evec // MatrixForm (* Column form *)
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} -\frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} & 0 \\ \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

In[ ]:=
```
evec[[1]]
```

Out[ ]= $\left\{-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0\right\}$

In[ ]:=
```
eval = Eigenvalues[mat2] ;
eval // Simplify // TableForm
```

Out[ ]//TableForm=

$\frac{4\pi}{3}$

$\frac{44\pi}{35}$

$\frac{32\pi}{35}$

In[ ]:= $\frac{\text{eval}}{\frac{4}{3}\pi}$ // Simplify

Out[ ]= $\left\{1, \frac{33}{35}, \frac{24}{35}\right\}$

In[ ]:= $\frac{\text{eval}}{\pi}$ // N

Out[ ]= {1.33333 , 1.25714 , 0.914286}

In[ ]:=
```
Transpose[evec] .mat2. evec // Simplify // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} \frac{4\pi}{3} & 0 & 0 \\ 0 & \frac{44\pi}{35} & 0 \\ 0 & 0 & \frac{32\pi}{35} \end{pmatrix}$$

In[ ]:=
```
Transpose[evec] . evec // Simplify // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Problem2:

2) Three equal mass points are located at (a,0,0), (0, a, 2a) and (0, 2a, a). Find the principal moments of inertia about the origin and a set of principal axes.

In[ ]:=
```
Clear["Global`*"]
```

Three equal mass points are located at (a,0,0), (0, a, 2a) and (0, 2a, a). Find the principal moments of inertia about the origin and a set of principal axes.

In[ ]:=
```
location = {
   {a, 0, 0},
   {0, a, 2 a},
   {0, 2 a, a}};
mass = {1, 1, 1};
v = {vx, vy, vz} = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
```

Note, we program diagonal and off - diagonal with different formulas below!!!

In[ ]:=
```
Clear[term]
term[n_, i_, i_] := +mass[[n]] (location[[n]].location[[n]] - (location[[n]].v[[i]])^2)
term[n_, i_, j_] := -mass[[n]] (location[[n]].v[[i]] * location[[n]].v[[j]])
```

In[ ]:=
```
mat = Table[ Sum[ term[n, i, j], {n, 1, 3}], {i, 1, 3}, {j, 1, 3}];
mat // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 10\, a^2 & 0 & 0 \\ 0 & 6\, a^2 & -4\, a^2 \\ 0 & -4\, a^2 & 6\, a^2 \end{pmatrix}$$

In[ ]:=
```
evec = (Normalize /@ Eigenvectors[mat]) // Simplify // Transpose ;
evec // MatrixForm (* Column form *)
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 0 & 1 & 0 \\ -\frac{1}{\sqrt{2}} & 0 & \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} & 0 & \frac{1}{\sqrt{2}} \end{pmatrix}$$

In[ ]:=
```
evec[[2]]
```

Out[ ]= $\left\{-\frac{1}{\sqrt{2}}, 0, \frac{1}{\sqrt{2}}\right\}$

In[ ]:=
```
eval = Eigenvalues[mat];
eval // Simplify // TableForm
```

Out[ ]//TableForm=

$10\,a^2$

$10\,a^2$

$2\,a^2$

In[ ]:=
```
Transpose[evec].mat.evec // Simplify // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 10\,a^2 & 0 & 0 \\ 0 & 10\,a^2 & 0 \\ 0 & 0 & 2\,a^2 \end{pmatrix}$$

In[ ]:=
```
Transpose[evec].evec // Simplify // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Problem3:

3)

10.35 ★★ A rigid body consists of three masses fastened as follows: m at $(a, 0, 0)$, $2m$ at $(0, a, a)$, and $3m$ at $(0, a, -a)$. **(a)** Find the inertia tensor **I**. **(b)** Find the principal moments and a set of orthogonal principal axes.

In[ ]:=
```
Clear["Global`*"]
```

Three equal mass points are located at (a,0,0), (0, a, 2a) and (0, 2a, a). Find the principal moments of inertia about the origin and a set of principal axes.

In[ ]:=
```
location = {
   {a, 0, 0},
   {0, a, a},
   {0, a, -a}};
mass = {1, 2, 3};
v = {vx, vy, vz} = {{1, 0, 0}, {0, 1, 0}, {0, 0, 1}};
```

Note, we program diagonal and off - diagonal with different formulas below!!!

In[ ]:=
```
Clear[term]
term[n_, i_, i_] := +mass[[n]] (location[[n]].location[[n]] - (location[[n]].v[[i]])^2)
term[n_, i_, j_] := -mass[[n]] (location[[n]].v[[i]] * location[[n]].v[[j]])
```

In[ ]:=
```
mat = Table[ Sum[ term[n, i, j], {n, 1, 3}], {i, 1, 3}, {j, 1, 3}];
mat // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 10\,a^2 & 0 & 0 \\ 0 & 6\,a^2 & a^2 \\ 0 & a^2 & 6\,a^2 \end{pmatrix}$$

In[ ]:=
```
evec = (Normalize /@ Eigenvectors[mat]) // Simplify // Transpose ;
evec // MatrixForm (* Column form *)
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & \frac{1}{\sqrt{2}} & -\frac{1}{\sqrt{2}} \\ 0 & \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \end{pmatrix}$$

In[ ]:=
```
evec[[2]]
```

Out[ ]=

$$\left\{0, \frac{1}{\sqrt{2}}, -\frac{1}{\sqrt{2}}\right\}$$

In[ ]:=
```
eval = Eigenvalues[mat] ;
eval // Simplify // TableForm
```

Out[ ]//TableForm=

$10\,a^2$

$7\,a^2$

$5\,a^2$

In[ ]:=
```
Transpose[evec].mat.evec // Simplify // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 10\,a^2 & 0 & 0 \\ 0 & 7\,a^2 & 0 \\ 0 & 0 & 5\,a^2 \end{pmatrix}$$

In[ ]:=
```
Transpose[evec] . evec // Simplify // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$